AUSTRALIAN ATOMIC ENERGY COMMISSION

RESEARCH ESTABLISHMENT

LUCAS HEIGHTS

A 1620 FORTRAN PROGRAMME FOR THE CALCULATION OF BRAGG ANGLES FOR USE IN POWDER X-RAY DIFFRACTION INVESTIGATIONS

by

D. G. WALKER

ABSTRACT

An I.B.M. 1620 programme is described which calculates the Bragg angles for cubic, hexagonal, and tetragonal crystal systems for use with powder X-ray diffraction techniques. Allowance is made for absent reflections in the case of face - and body - centred cubic and close-packed hexagonal materials.

1. GENERAL CONSIDERATIONS

The existence of tabulated data giving the Bragg angles for various wavelengths allows much time to be saved in extensive powder X-ray diffraction investigations on a particular material. For example, it allows rapid selection of wavelength to give optimum line positions, rapid indexing of patterns and rapid setting of a diffractometer when only a few lines are to be scanned. All that is required for the calculations are values of the lattice parameters, which may be more or less accurate depending on the application of the resulting data. However, such calculations are too tedious and time-consuming except where computer facilities are available.

Most powder investigations are restricted to unit cells having not more than two parameters (cubic, hexagonal, and tetragonal) and the programme described has been limited to these cases. If necessary it could be extended to deal with more complex crystal systems.

The programme calculates the Bragg angles of all reflecting planes of the form (h, k, l) once only up to the Bragg cut-off. In the simple cubic system, this condition may be expressed as:

$$h \geq k \geq l \quad \text{for } \sin\theta \leq 1,$$

and for the simple hexagonal and tetragonal systems as:

$$h \geq k \quad \text{for } \sin\theta \leq 1.$$

Further restrictions arise in the case of lattices exhibiting characteristic absences. In general, the computation can be carried through for the simple lattice, and the absent reflection deleted subsequently by the user. However, two special cases, the face - and body - centred cubic structures, warrant further attention. In these cases, the number of possible reflections is substantially less than for a simple cubic structure of the same parameter (approximately one third for the face-centred cubic), so that there is considerable wastage of machine time if the calculation is made for the primitive cell. For this reason the programme has been arranged to allow the selection of the conditions:

h, k, and l either all odd or all even, and

h + k + l even,

for the face - and body - centred structures respectively. Although less important in terms of machine time, the condition:

h = k only for l even

has been included to deal with close packed hexagonal structures, since these arise frequently in work done by the Materials Physics Section.

The logic used in this programme is illustrated in Figure 1 which shows a flow diagram for the computation of all possible reflections for a face-centred cubic lattice. Similar logic and arrangement are used for other cases.

2. PROGRAMME DETAILS

The programme is written in Fortran I with Format and is listed in Appendix 1. Reference to this will show that the first data input contains, in sequence, the number of wavelengths to be computed, the crystal system index, the characteristic absences index and the "a" and "c" parameters. The first three of these are in fixed point and have been allotted fields of 3, 2, and 2 respectively. The coding for the indices is as follows:

(a) Crystal System Index

1 = Cubic; 2 = hexagonal; 3 = tetragonal

(b) Characteristic Absences Index

Reflections only when:

1 = h, k, l all odd or all even; 2 = h + k + l even
3 = (h=k) only for l even; 4 = all h, k, l.

The parameters are in floating point and have each been allotted a field of 9 with 5 places of decimals. In the case of cubic crystals, any value (or blanks) may be used for the "c" parameter. After reading, the parameters are printed and punched for reference. The various wavelengths are then read in on separate cards, each having a field of 8 with 5 decimal places. For example, consider the calculation of reflections from a body centred cubic structure, "a" = 2.652, for copper $K\alpha$, and $K\beta$ radiation. The data would be as follows, where "b" is used to indicate blanks:

Card 1 bb2b1b2bb2.625bbbb0.
Card 2 b1.54050
Card 3 b1.39217

The Bragg angles for each wavelength are computed in a "DO" loop and at the start of each loop the corresponding wavelength is printed and punched. Fixed point arithmetic is used to decide between odd and even values where characteristic absences are involved. The output consists of the Bragg angles (in degrees of θ) terminated at 3 decimal places together with the corresponding Miller indices, arranged in columns and headed, respectively, H, K, L, THETA. Sense switch 1 is used to give the output as punched (off) or printed and punched (on). After the values for all the input wavelengths have been computed, the instruction "FINISH LOAD NEXT DATA" is printed on the typewriter. Further calculations can then be initiated by placing the corresponding data cards in the reader hopper and pressing "Reader Start".

An average value of the speed of computation for this programme using the I.B.M. 1620 computer with punched output is about 25 reflections per minute.

APPENDIX 1

PROGRAMME LISTING

```
      DIMENSION WAV(50),SN(50),THETA(50)
    1 READ 1000,N,NS,NCA,A,C
      PRINT 1001,A,C
      PUNCH 1001,A,C
      DO 27 I=1,N
      READ 1002,WAV(I)
      PRINT 1003,WAV(I)
      PUNCH 1003,WAV(I)
      IF(SENSE SWITCH 1)5,6
    5 PRINT 1006
    6 PUNCH 1006
      J=1
      K=0
      L=0
    2 GO TO(3,8,9,11),NCA
    3 JK = J-K
      KL = K-L
      JK2 = JK/2
      KL2 = KL/2
      IF(JK-2*JK2) 28,28,29
   28 IF(KL-2*KL2) 11,11,21
   29 IF(KL-2*KL2) 19,19,21
    8 JS=J+K+L
      JS2=JS/2
      IF(JS-2*JS2)19,11,19
    9 IF(J-K)11,10,11
   10 L2=L/2
      IF(L-2*L2)19,11,19
   11 GO TO(12,13,14),NS
   12 XJKL=(J*J)+(K*K)+(L*L)
      SN(I)=WAV(I)*SQRT(XJKL)/(2.*A)
      GO TO 16
   13 YJK=(J*J)+(J*K)+(K*K)
      XJK=4./3.*YJK
      GO TO 15
   14 XJK=(J*J)+(K*K)
   15 XL=L*L
      SN(I)=WAV(I)/2.*SQRT((XJK/(A*A))+(XL/(C*C)))
   16 IF(SN(1)-1.)17,17,20
   17 THETA(I)=57.29578*ATAN(SQRT((SN(I)*SN(I))/(1.-(SN(I)*SN(I)))))
      IF(SENSE SWITCH 1)18,7
   18 PRINT 1005, J, K, L, THETA(I)
    7 PUNCH 1005, J, K, L, THETA(I)
   19 J=J+1
      GO TO 2
   20 IF(K+1-J)21,22,22
   21 K=K+1
      J=K
      GO TO 2
```

APPENDIX 1 (continued)

```
   22 GO TO(23,25,25),NS
   23 IF(L-K)24,27,27
   24 L=L+1
      K=L
      J=K
      GO TO 2
   25 IF(K)27,27,26
   26 L=L+1
      K=0

      J=0
      GO TO 2
   27 CONTINUE
      PRINT 1004
      GO TO 1
 1000 FORMAT (I3,I2,I2,F9.5,F9.5)
 1001 FORMAT (11H PARAMETERS,5X F9.5,5X F9.5)
 1002 FORMAT (F8.5)
 1003 FORMAT (11H WAVELENGTH,2X F8.5)
 1004 FORMAT (7H FINISH,5X 15H LOAD NEXT DATA)
 1005 FORMAT (I3,2X I3,2X I3,14X F7.3)
 1006 FORMAT (3H  H,2X 3H  K,2X 3H  L,15X 5HTHETA)
      END
```

END



FIGURE 1

BLOCK DIAGRAM OF SCHEME OF COMPUTATION FOR A FACE-CENTRED CUBIC MATERIAL